
 § 1. НАЧАЛО РАБОТЫ С ЯЗЫКОМ БЕЙСИК.  ОСНОВНЫЕ ПОНЯТИЯ

После того, как вы включили компьютер, на экране появятся две синие таблицы. Это - так называемые таблицы Нортон-коммандера, названные по имени их автора. Как работает Нортон-коммандер, можно разобрать отдельно,  сейчас ограничимся основными простыми понятиями. В таблице Нортона по сути приведен каталог главных разделов (директорий), имеющихся на жестком диске компьютера. В каждом разделе может находиться несколько подразделов, и в каждом подразделе находится определенное количество компьютерных документов - файлов. В общем, все устроено точно так же, как в каталоге любой библиотеки. Чтобы начать вашу самостоятельную работу с компьютером, создадим сначала каталог, в котором будете работать лично вы. Для этого нажмите клавишу F7 из верхнего ряда, на экране появится табличка с надписью create the directory (создать директорию), наберите grup35, если вы учитесь, предположим, в 35 группе, после этого нажмите клавишу Enter. В таблице перед вами появится новая директория GRUP35. Нажимая клавиши со стрелками , переместите курсор (подсвеченное название) на эту директорию и нажмите Enter. При этом вы войдете  внутрь директории, в которой пока ничего не содержится - перед вами горит только курсор. Создадим теперь здесь поддиректорию с вашей фамилией, нажмите F7 и наберите первые несколько букв вашей фамилии (латинскими буквами), например fedot, если фамилия Федотов. При этом появится поддиректория FEDOT, переместите курсор на нее и нажмите Enter (войдите в нее).

Теперь вы находитесь внутри вашего личного каталога и можете работать с нужными вам программами, не влияя на работу в других директориях. Поскольку вы хотите работать с языком программирования Бейсик, вам нужно создать внутри вашего каталога поддиректорию Бейсик. Для этого нажмите еще раз F7 и наберите basicf, нажмите Enter , перед вами появится BASICF (мы добавили первую букву вашей фамилии, чтобы не возникало путаницы). Теперь вам осталось выполнить еще одну важную операцию - записать в поддиректорию BASICF файл, запускающий работу программы с языком Бейсик. Этот файл выглядит так gwbasic.exe и находится в каталоге (директории) BASIC. Итак, войдите в вашу поддиректорию BASICF, нажмите клавишу Tab, при этом курсор переместится на правую таблицу, найдите в этой таблице директорию BASIC, войдите в нее, найдите внутри файл gwbasic.exe, переместите курсор на него и нажмите клавишу F5 в верхнем ряду. Перед вами появится табличка с надписью  Copy  gwbasic.exe to C:\GRUP35\FEDOT\BASICF то есть копирую gwbasic.exe в... и указан путь перехода из каталога в подкаталог и так далее в указанное вами место, в этой команде C:\ означает название главного каталога (корневого), находящегося на жестком диске (диске С). Нажмите Enter и внутри поддиректории BASICF возникнет файл gwbasic.exe. Вот теперь вы, наконец, можете начать работу с языком программирования Бейсик. Запустите файл gwbasic.exe, то есть переместите на него курсор и нажмите Enter. Экран компьютера станет темным и в левом верхнем углу появится OK, это значит, что язык Бейсик загружен в оперативную память компьютера и можно начать изучение этого языка программирования.

Язык Бейсик это один из наиболее простых языков общения с персональными ЭВМ. Ваш "разговор" с ПЭВМ на языке Бейсик состоит в том, что вы приказываете ПЭВМ выполнить то или иное действие, употребляя для этой цели определенные команды. В качестве команд используется небольшое количество слов на английском языке. Для знакомства с тем, как это происходит, рассмотрим ряд простых примеров.

Сосчитать:



 EMBED Equation.2  
                                         (1)

Набираем на клавиатуре:

 PRINT  (3+4)*6/(25-4)

Разъясним это выражение, написанное на языке Бейсик. . Слово означает ПЕЧАТАЙ. Мы видим, что знак умножения выглядит как * , знак деления это косая черта /, написание же знаков сложения и вычитания не отличается от общепринятых. Обратим внимание, что на языке Бейсик все пишется в одну строчку.     Итак, мы набрали строчку (команду) и на экране ПЭВМ также появилась знак за знаком эта строчка. Нажимаем теперь клавишу Enter ( и на экране образуется:

 PRINT (3+4)*6/(25-4)

 2                                                                         (2)

 OK

Обозначение OK является сигналом от ПЭВМ, что она выполнила ваше задание и ждет следующих указаний. Как легко проверить, результат решения примера действительно двойка. Команда  PRINT  (ПЕЧАТАЙ)  может быть использована и для печати на экране ПЭВМ какого-либо текста.  Например, мы дадим следующую команду компьютеру: 

 PRINT "Результат вычислений равен -", (3+4)*6/(25-4) 

На экране образуется:

Результат вычислений равен -         2              (3)

OK

Итак, для печати текста, мы видим, нужно использовать  открывающие и закрывающие скобки. Таким образом, мы уже научились формировать осмысленные выражения на экране ПЭВМ.

 
То, чем мы занимались до сих пор, называется режимом непосредственного исполнения. В этом режиме ПЭВМ выполняет ваше задание сразу же после его формирования. Однако, так можно делать, если задания короткие, небольшие. В случае более сложных заданий используется другой режим работы ПЭВМ - режим составления программ. Изучим особенности этого режима. Для начала возьмем самые простые задания.

Пусть наша программа выглядит так:

 10 LET A=3           

 20 LET B=4           

 30 LET C=A+B                                              (4)

 40 PRINT A,B,C  
 50 END                 
Слово LET означает по-русски ПУСТЬ, слово END - КОНЕЦ. Слово LET в языке Бейсик означает оператор присваивания. Таким образом, LET A=3 означает - присвоить величине А значение, равное числу 3. Что такое операция присваивания, мы уже знаем [1,2]. Но раньше мы написали бы а:=3, а весь алгоритм нашего задания мы написали бы в виде:

 1 a:=3

 2 b:=4

 3 c:=a+b                                                             (5)

 4 печатать a,b,c

 5 конец
Сравнивая (4) и (5), мы видим, что в командах языка Бейсик используются только заглавные буквы латинского алфавита. Каждая строка программы имеет свой номер,  обычно принято писать в программах номера с интервалом 10. Это позволяет при необходимости вставить между строками еще некоторое количество новых строк с промежуточными номерами (если возникает необходимость).

При наборе программы на клавиатуре одновременно весь текст программы строка за строкой появляется на экране. Как же теперь заставить ПЭВМ выполнить программу? Для этого    существует специальная команда - так называемая директива RUN -в переводе с английского  (БЕГИ, ГОНИ). Таким образом, окончательно:

 10 LET A=3

 20 LET B=4

 30 LET C=A+B                                           (6) 

 40 PRINT A,B,C

 50 END

 RUN              
 3    4    7

 OK

Однако, все будет так, только если мы не допустили ошибок при наборе программы, а такие ошибки вначале неизбежно будут. Как быть с такими ошибками? Для этого существуют специальные приемы. Во-первых, клавиша "забой"   (расположена крайней справа в верхнем ряду, там, где цифры). Эта клавиша работает в пределах одной текущей строки. Таким образом, если вы, например, набрали 20 LET BB=, то для того, чтобы исправить ошибку, нужно нажать на клавишу   2 раза (в этом случае), тогда три последних знака зачеркнутся и можно будет продолжить набор строки уже правильно, т.е. продолжить B=4  .

Если же вы уже набрали 20 LET BB=4  , то вы уже перешли на новую строку и клавиша   вам не поможет. Но выход есть - нужно снова набрать строку 20, но постараться сделать это без ошибок.

 В результате на экране будет:

 20 LET BB=4

 20 LET B=4                                             (7)

Предположим, что при наборе программы (10) оказалось, что вы забыли набрать строку 40. Ничего страшного, наберите ее вслед за последней строкой программы. А теперь наведем порядок в набранной нами программе. Для этого служит специальная команда - директива LIST (по-английски - СПИСОК). Наберите это слово с новой строки, затем   и на экране появится отредактированная программа уже в нужном нам виде:

 10 LET A=3                  LIST 
 20 LET BB=4               10 LET A=3

 20 LET B=4                  20 LET B=4

 30 LET C=A+B            30 LET C=A+B

 50 END                         40 PRINT A,B,C                         (8)

 40 PRINT A,B,C           50 END

Таким образом, автоматически неправильная строка 20 заменена верной строкой, и автоматически строки расположились в порядке возрастания номеров - директива LIST отредактировала программу. Остается только набрать RUN   и ПЭВМ исполнит программу.

Существует еще один способ исправления (редактирования) уже набранной строки - использование специальной команды EDIT (редактируй). Нужно набрать EDIT 20 и на экране появится строка 20 с курсором внутри нее, передвигаем курсор в те места, которые нужно исправить, вносим исправления и нажимаем . Этот прием удобно применять и при наборе не слишком отличающихся друг от друга строк программы. Например, в программе (12) строки 20 и 30 можно создать, просто два раза отредактировав набранную уже строку 10, при этом изменяя номер строки и внося нужные изменения. Ясно, что в случае набора длинной программы с мало отличающимися строками будет большая экономия времени.

Пусть теперь мы хотим выполнить какую-либо другую программу. Перед ее набором неплохо уничтожить программу, с которой мы работали только что. Это можно сделать, нажав клавиши:

     10  20  30  40  50                                               (9)

Набрав после этого LIST, мы убедимся, что программа действительно исчезла. Таким образом мы вычеркнули всю программу строчку за строчкой. Существует и более быстрый прием вычеркивания ненужных строчек:   DELETE 10-30   -вычеркиваются строки с 10-й по 30-ю. Слово DELETE означает -СТИРАТЬ. Всю программу можно быстро вычеркнуть, убрать из оперативной памяти ПЭВМ, набрав NEW  (НОВАЯ).

Выполним теперь пример на умножение:

 100 LET F=5

 110 LET M=15

 120 LET P=F*M                                                             (10)

 130 PRINT F,M,P

 140 END

 RUN

 5    15    75

Если мы теперь заменим строку 130 на следующую:

 130 PRINT F;"умножить на";M;"равно";P

 то после команды RUN на экране появится более "человеческая" строчка:   5 умножить на 15 равно 75.


Обратим внимание на то, что использование знака  ;  в строке 130 приводит к печати результата на экране с интервалами в один пробел между числами и словами, если бы мы использовали в этих местах запятую, то интервал при печати был бы в семь пробелов - редкая печать. В рассмотренном примере она не подходит.

Обратимся теперь к примеру, уже знакомому нам по работе [3]:

найти по формуле  у = 

  значение "у" при заданном значении “х”. Пусть, например, х=9. Тогда можно написать два варианта решения этой задачи.

 10 LET X=9:LET A=2*X                      10 LET X=9

 20 LET B=A+3:LET C=3*X                 20 PRINT (2*X+3)/(3*X-4)

 30 LET D=C-4:LET Y=B/D                  30 END

 40 PRINT Y

 50 END                                                                                                 (11)

В этой задаче мы дополнительно встречаемся с новым знаком "двоеточие" - в языке Бейсик  это означает соединение двух, или более соседних строчек (команд) в одну строчку, при этом запись программы, конечно, оказывается более компактной.

Рассмотрим, как выглядят на языке Бейсик некоторые алгебраические выражения.

sin



 EMBED Equation.2  


 EMBED Equation.2  
                                          общепринятая запись

  SIN ((9.5-SQR(5))/(5.6+2.6))                 запись на языке Бейсик

a

 

                                                общепринятая запись

  A * B^3/C/D + E*F/LOG(X)                   запись на языке Бейсик

 
Итак, мы видим, что в языке Бейсик используются только заглавные буквы латинского алфавита, формулы пишутся в одну строку, корень квадратный пишется SQR, логарифм натуральный - LOG, при возведении в степень используется знак - ^.

§ 2. ОРГАНИЗАЦИЯ ДИАЛОГА С ПЭВМ

     Усложним задачу (11). Пусть нам нужно найти "у" при разных значениях "х", которые будем задавать произвольно. В этом случае программа выглядит так:

 10 INPUT X

 20 PRINT (2*X+3)/(3*X-4)

 30 END                                                                        (12)

Итак, мы используем новый оператор INPUT (ВВОД). Как он работает? После набора RUN на экране появляется знак вопроса:

RUN                                         (машина ждет, когда вы, по-вашему
 ?                                                  желанию назначите значение х)

 
Вы должны теперь ответить на вопрос машины - набрать на клавиатуре нужное число. Пусть это будет 2. Тогда ваш разговор с ПЭВМ будет выглядеть так:

 10 INPUT X

 20 PRINT (2*X+3)/(3*X-4)

 30 END

 RUN                                                                             (13)

 ?

 2   
 3.5

 OK

Итак, на экране появилось число 3.5 - это и есть значение "у". Пусть теперь мы хотим, чтобы программа могла по очереди вычислять значение "у" при разных значениях "х". Для этого изменим программу:

 10 INPUT X

 20 PRINT (2*X+3)/(3*X-4)                                       (14)

 30 GOTO 10

Команда GOTO 10 означает ИДИ К 10, т.е. перейти к строке 10. GOTO - это оператор безусловного перехода. Как будет теперь работать наша программа?

 10 INPUT X

 20 PRINT (2*x+3)/(3*X-4)

 30 GOTO 10

 40 END                                                                                   (15)

 RUN      
 ?                                        (ПЭВМ спрашивает значение "х")

 2        
 3.5

 ?                                         (ПЭВМ спрашивает новое значение "х")

 4        
 1.375

 ?             (и так далее)

Мы можем сделать наш разговор с ПЭВМ более "человеческим". Изменим для этого программу. Однако сначала нам нужно выйти из программы (15), иначе она будет без конца задавать вопросы о новых значениях "х". Нажмем клавишу Ctrl и, удерживая ее, нажмем клавишу Pause - на экране сразу же появится OK, можно продолжать работу. Итак, изменим программу (15):

 5  PRINT "введи значение х"

 10 INPUT X

 20 PRINT "у равно";(2*X+3)/(3*X-4)

 30 GOTO 10                                                                                    (16)

 40 END

Теперь диалог с машиной будет таким:

 RUN      
 введи значение х

 ?

 2        
 у равно 3.5

 введи значение х

 ?

 4       
 у равно 1.375

 введи значение х           (и так далее)

 ?

Остановить работу программы (16) можно, нажав две клавиши Ctrl Pause. Однако, можно поступить иначе, изменив программу так:

 10 INPUT X

 15 IF X=0 THEN 40

 20 PRINT (2*X+3)/(3*X-4)

 30 GOTO 10                                                                       (17)

 40 END

Мы ввели дополнительную строку 15, которая по-русски звучала бы так: ЕСЛИ Х=0 ТОГДА 40. Итак, мы используем новые для нас операторы языка Бейсик - IF, THEN - операторы условного перехода (то есть перехода по какому-то условию). В нашем случае таким условием является равенство Х=0, если мы, когда нам надоест работать с программой, в ответ на вопрос машины (INPUT X) введем ноль, то осуществится переход по адресу 40, и программа прекратит работу.

Обратим внимание на то, что внизу экрана имеются надписи  1 LIST  2 RUN  3 LOAD(   4  SAVE ( и так далее, из них для нас важны первые четыре. Первые две означают, что известные вам команды LIST и RUN можно быстро вызвать, нажимая клавиши F1 и F2. Четвертая запись позволяет сохранить (по английски  SAVE) любую программу из тех, что вы набираете. Например, вы работаете с программой 17 и хотите сохранить ее, для этого нужно нажать клавишу F4 и после появления на экране команды  SAVE ( набрать проивольное имя программы, например fed17 (обязательно латинскими буквами), и нажать ( . Теперь в любой момент вы сможете вызвать эту программу для работы с ней, даже на следующем занятии. Для этого нужно нажать F3, на экране появится LOAD( (по-английски - загрузить), набрать имя fed17, нажать ( и перед вами появится программа 17. Если вы к следующему занятию забыли это имя, наберите команду files (файлы), перед вами появится список имеющихся в вашей Бейсиковской директории файлов и вы сможете увидеть имя программы, которую вы хотите загрузить.

§ 3. ПРОГРАММЫ ДЛЯ ЦИКЛИЧЕСКИХ АЛГОРИТМОВ

Попробуем задать ПЭВМ следующую задачу - надо вычислить квадраты целых чисел от 1 до 20 и напечатать на экране сами числа и их квадраты. Изобразим блок-схему алгоритма этой задачи:

                                                                                Рис. 1

А теперь напишем программу на языке Бейсик, соответствующую этому алгоритму.

 10 LET N=1                                                  пусть N=1

 20 IF N>20 THEN 70                                  если N>20  тогда 70

 30 LET S=N^2                                              пусть S  равно N в квадрате  

 40 PRINT N,S                                               печатай N и S

 50 LET N=N+1                                             пусть N=N+1

 60 GOTO 20                                                  иди к 20                                            (18)

 70 END                                                          конец      

Мы видим, что программа и блок-схема шаг за шагом в точности соответствуют друг другу. Обратим внимание, что N в квадрате на языке Бейсик - это N^2. Если мы теперь запустим программу, то на экране мы увидим:

RUN

 1      1

 2      4

 3      9

 .      .                                                                         (19)

и так далее

 .      .

20     400

 OK

Таким образом, мы организовали исполнение циклического алгоритма, используя операторы условного (IF ....THEN) и безусловного (GOTO) переходов. Однако в языках программирования и соответственно в Бейсике существует специальный, более простой способ для осуществления циклических программ.

Запишем снова нашу программу, но уже в новом виде:

 10 FOR N=1 TO 20                   для N от 1 до  20

 20 PRINT N,N^2                        печатай N и N2                 (20)

 30 NEXT N                                 cледующее N
 40 END                                        конец
Итак, FOR, TO, NEXT - это ключевые слова операции организации цикла. Программы типа (18) и (20) могут быть использованы для создания циклов, не имеющих никакого отношения к вычислениям. Например, следующая программа (в двух способах написания):

10 LET N=1

 20 IF N>30 THEN 60

 30 PRINT "желаю успехов"

 40 LET N=N+1                                                               (21)

 50 GOTO 20

 60 END

 10 FOR N=1 TO 30

 20 PRINT "желаю успехов"

 30 NEXT N                                                                     (22)

 40 END
Эти программы печатают 30 раз фразу "желаю успехов". Обратим внимание, что число N играет теперь просто роль счетчика. Продемонстрируем теперь возможности ПЭВМ в выполнении быстрых и довольно сложных вычислений.

Требуется решить такой пример из [5]:

 X6 +46 +86 +86 +36 +46 =X48834                                   (23)

Надо найти такое число "X" в интервале 1-9, которое было бы решением примера (23).  Решить этот пример, даже пользуясь микрокалькулятором, довольно сложно. Для ПЭВМ же достаточно написать простую программу и ответ будет получен через несколько секунд при ее исполнении.

Итак, эта программа и ее исполнение:

 10 FOR X=1 TO 9

 20 PRINT X^6+4^6+8^6+8^6+3^6+4^6,

 30 PRINT X*100000+48834

 40 NEXT X                                                                    (24)

 50 END
     Запустим программу:

RUN
 533210       148834

 533237       248834

 533938       348834

 537305       448834

 548834       548834                                 Вот   решение: Х=5

 579865       648834

 650858       748834

 795353       848834

 1064650      948834

 OK
Эти две колонки цифр возникают практически мгновенно - вот каковы возможности ПЭВМ (и хорошей программы для решения). Рассмотрим теперь пример, который покажет, что хотя ЭВМ     считает быстрее, хороший ученик все равно остается умнее ее.

Задача - сосчитать сумму целых чисел от 1 до 100. Машина решит эту задачу с помощью программы:

 10 LET S=0

 20 FOR N=1 TO 100

 30 LET S=S+N

 40 NEXT N                                                              (25)

 50 PRINT S
 60 END
 RUN
 5050

Таким образом, машина просто сложит все числа одно за другим. Умный же ученик поступит так:

          100

         1+99=100

         2+98=100

         3+97=100               100+49х100+50=5050

         . .  .

        47+53=100                                                               (26)

        48+52=100

        49+51=100

        50

                 § 4 ШКОЛЬНАЯ ТАБЛИЦА УМНОЖЕНИЯ

Попробуем теперь использовать операторы организации цикла для печати на экране ПЭВМ школьной таблицы умножения.

 100 LET A=3

 110 FOR B=1 TO 5

 120 PRINT A,B,A*B

 130 NEXT B

 140 END
 RUN                                                                       (27)

 3       1       3

 3       2       6

 3       3       9

 3       4       12

 3       5       15

Программа проста и не требует пояснений. Теперь усложним ее.

 100 FOR A=3 TO 4

 110 FOR B=1 TO 5

 120 PRINT A,B,A*B                                                 (28)

 130 NEXT B

 140 NEXT A

 150 END

 RUN

 3       1       3                                умножить на 3

 3       2       6                                числа от 1 до 5

 3       3       9

 3       4       12

 3       5       15 

 4       1       4                                умножить на 4

 4       2       8                                числа от 1 до 5

 4       3       12

 4       4       16

 4       5       20

Программа (28) является двойным циклом, в ней имеется внутренний цикл - изменение В от 1 до 5, и внешний цикл - изменение А от 3 до 4. Сравнивая (28) и (27), мы видим, что программа (28) является естественным продолжением программы (27). Если в программе (27) мы меняли только значение второго сомножителя, то в программе (28) мы меняем и значение первого сомножителя.    В результате получаем на экране ПЭВМ часть таблицы умножения. Чтобы получить всю таблицу умножения, сначала поступим так:

 100 FOR A=1 TO 10

 110 FOR B=1 TO 5

 120 PRINT A,B,A*B,

 130 NEXT B                                                        (29)

 140 NEXT A
 150 END
При выполнении программы (29) мы получим только половину таблицы умножения. Совершим, наконец, последнее изменение:

 100 FOR A=1 TO 10

 110 FOR B=1 TO 10

 120 PRINT A,B,A*B,

 130 NEXT B                                                        (30)

 140 NEXT A
 150 END
Эта программа уже напечатает на экране строчка за строчкой всю таблицу умножения. Конечно ,такое изображение неудобно для рассматривания. Мы можем улучшить дело и добиться того, чтобы на экране возникла таблица умножения именно в том виде, как она выглядит на обложке школьной тетради. Итак, программа:

10 FOR A=2 TO 9

 20 FOR B=2 TO 5

 30 PRINT B;"x";A;"=";A*B,

 40 NEXT B

 50 NEXT A

 55 PRINT                                                                      (31)

 60 FOR A=2 TO 9

 70 FOR B=6 TO 9

 80 PRINT B;"x";A;"=";A*B,

 90 NEXT B

 100 NEXT A
 110 END
При исполнении этой программы на экране появится вся таблица умножения, но разбитая на две части, как это и делается на обложке школьной тетради.

Разъясним теперь действие строчек 30 и 80. Точка с запятой, как мы помним, означает, что следующий знак нужно печатать близко от предыдущего, а вот запятая означает, что надо отступить на некоторое число позиций - как раз, чтобы перейти к печати соседней колонки таблицы умножения; "х" - означает печать на экране знака умножения; "=" - означает печатание знака равенства.

Команда 55 означает, что нужно печатать - что? - пустую строку, то есть ничего не печатать, пропустить одну строку. Эта пустая строка как раз и имеется в школьной таблице умножения, она разделяет верхнюю и нижнюю половины таблицы.

            § 5 ПРИМЕРЫ ИСПОЛЬЗОВАНИЯ ПЭВМ НА УРОКАХ

Познакомимся еще с одной программой и ее исполнением.

 10 PRINT "Вычисление суммы чисел"

 20 DATA 3,6

 30 READ C,D

 40 LET S=C+D

 50 PRINT C,D,"сумма=";S

 60 END                                                                         (32)

 RUN
 Вычисление суммы чисел

 3       6       сумма=9

Разъясним работу программы. Мы используем новые операторы - операторы ввода данных: DATA - ДАННЫЕ, READ - ЧИТАЙ. Команды 20 и 30 означают то же самое, как если было бы:

 20 LET C=3

 30 LET D=6

Теперь пусть мы хотим, воспользовавшись нашими новыми знаниями, напечатать заголовок какого-либо документа. Например, пусть мы начали нашу программу так:

 10 PRINT "20.06.2000"

 20 PRINT "ТАБЕЛЬ УСПЕВАЕМОСТИ"

При исполнении программы на экране появится в левом верхнем углу:

 20.06.2000

 ТАБЕЛЬ УСПЕВАЕМОСТИ

Однако, это нам не нравится. Мы хотели бы, чтобы дата оказалась в правом углу, а заголовок - на середине строки. Можно так сделать, для этого используем программу:

 10 PRINT TAB(50)"20.06.2000"

 20 PRINT TAB(20)"ТАБЕЛЬ УСПЕВАЕМОСТИ"

 и тогда на экране получится то, что мы и хотели. Это произойдет потому, что мы использовали новый оператор PRINT TAB. Команда PRINT TAB(50) означает - печатай, отступя 50 позиций от левого края. Покажем, как можно применить этот оператор для конкретной задачи - построения графика зависимости, допустим, успеваемости за пять лет по годам.

Пусть мы имеем следующий набор чисел: 30,20,10,24,40 - это количества неуспевающих учеников по годам пятилетки. Принято говорить, что мы имеем так называемый информационный массив чисел N(1)=30, N(2)=20, N(3)=10, N(4)=24, N(5)=40. Рассмотрим теперь работу следующей программы:

 10 DIM N(5)                                   задается длина массива

 20 DATA 30,20,10,24,40              задаются числа массива

 30 FOR I=1 TO 5                            заголовок цикла                                        (33)

 40 READ N(I)                                 читай очередное число

 50 PRINT TAB(N(I))"*"                  печатай * на расстоянии N(I)

 60 NEXT I                                       следующее I - следующий год

 70 END
Команда 10 использует новый оператор DIM - в переводе с английского (DIMENSION) это означает - размер, величина. Итак DIM N(5) означает, что наш информационный массив содержит пять чисел (строго говоря, шесть чисел, так как может быть и число N(0)). Остальные строки программы объяснены справа от программы. При выполнении программа (33) напечатает по сути график изменения по годам количества неуспевающих учеников. Конечно, неудобно, что график повернут на бок, однако если у нас есть возможность работать с принтером, то есть переносить текст с экрана ПЭВМ на бумагу, тогда уже нетрудно листок бумаги повернуть в нужное положение.

Рассмотрим еще один пример использования оператора PRINT TAB для рисования графиков.

 10 FOR X=1 TO 8

 20 LET Y=X^2

 30 PRINT TAB(Y)"*"

 40 NEXT X                                                       (34)

 50 END
При выполнении этой программы на экране появится зависимость Y=X2 ,- парабола. Такой прием позволяет изображать на экране графики и других зависимостей, для этого нужно изменить написание строки 20.

Познакомимся еще с одним оператором (вернее функцией) языка Бейсик. Составим программу, имитирующую бросание игральной кости (кубик с шестью гранями, на которых изображены от одной до шести точек).

 10 LET X=RND(1)*6+1

 20 LET A=INT(X)

 30 PRINT A                                                      (35)

 40 END
В строке 10 появилось новое - функция RND(1) -ее значение равно любому случайному числу (десятичной дроби) в интервале от 0 до 1. RND(1)*6 -умножение на 6 приводит к тому, что интервал увеличится от 0 до числа 5,.....- какое-то число знаков после запятой. Тогда RND(1)*6+1 изменит интервал, числа будут от 1 до 6,....... В строке 20 используется еще одна функция -INT(X) -это означает целую часть от десятичного числа. То есть, если, например, Х=3,1423, то INT(X)=3, знаки после запятой отбрасываются.

Итак, LET A=INT(X) в нашем случае будет означать, что число А будет принимать значения целых чисел, лежащих в интервале 1-6. И вся наша программа будет при каждом ее исполнении печатать на экране произвольно любое целое число в интервале 1-6, то есть будет имитировать бросание игральной кости.

Теперь мы сможем составить простую программу, которая данное "бросание костей" использует для контроля знаний ученика. Имеется в виду умение перемножать числа.

Вот эта программа и ее возможная работа:

10 LET A=INT(RND(1)*10+1)                                     (любое число из 1-10)

 20 LET B=INT(RND(1)*20+1)                                     (любое число из 1-20)

 30 LET C=A*B

 40 PRINT"Сколько получится, если перемножить числа"

 50 PRINT A,B

 60 INPUT X

 70 IF X=C THEN PRINT"Правильно":GOTO 10

 80 PRINT"Неверно":GOTO 10

 90 END                                                                                                            (36)

 RUN
 Сколько получится, если перемножить числа

 6      17

 ?                                                                                       (Вы должны ответить)

 102

 Правильно

 Сколько получится, если перемножить числа

 3      8

 ?

 26

 Неверно                                                                            (и так далее)

Таким образом, ПЭВМ каждый раз выбирает два произвольных числа А и В, перемножает их и предлагает вам ответить, каков будет этот результат. Обратим внимание, что в строках 70 и 80  используется действие знака "двоеточие" - в языке Бейсик это означает разделение двух соседних команд. В строке 80 хочется написать в ее начале IF X<>C THEN....(Х не равно С), однако, немного подумав, мы поймем, что это уже были бы лишние слова.

Дадим теперь примеры использования языка Бейсик на уроках иностранного языка. Первая программа (проверка знания английских слов):

10 LET C=1

 20 ON C GOTO 31,32,33,34,35,36,37

 31 PRINT "Mother":LET A=3:GOTO 40

 32 PRINT "Father":LET A=2:GOTO 40

 33 PRINT "Sister":LET A=1:GOTO 40

 34 PRINT "Brother":LET A=5:GOTO 40

 35 PRINT "Daughter":LET A=4:GOTO 40

 36 PRINT "Son":LET A=6:GOTO 40                                             (37)

 37 GOTO 90

 40 PRINT "Введите номер нужного русского слова"

 50 PRINT "Брат 3, Дочь 4, Мать 5"

 55 PRINT "Отец 6, Сестра 7, Сын 2"

 60 INPUT B

 70 IF A+B=8 THEN PRINT "Верно":LET C=C+1:GOTO 20

 80 PRINT "Неправильно":LET C=C+1:GOTO 20

 90 END
Разберем, как работает эта программа. Здесь мы применили еще один новый оператор языка Бейсик - ON - так называемый оператор-переключатель, в нашей программе при изменении С от 1 до 7 происходит переход из строки 20 в строки 31-37 соответственно. При каждом прохождении программы С увеличивается на единицу согласно строкам 70 и 80. Пусть, например, С=4. Тогда выполняется строка 34 - на экране появляется слово Brother, переменной величине А присваивается значение 5 (этого ученик не узнает), и происходит переход к строке 40, то есть ПЭВМ предлагает из шести русских слов выбрать правильное слово. Если вы выберете слово Брат (число 3), то согласно строке 70 на экране появится "Верно" и С станет равным 5. Если же вы ошиблись, то условие А+В=8 не выполнится, на экране появится "Неправильно", С  станет равным 5, на экране появится следующее английское слово.

Программа будет "вежливой", если мы изменим строку 90 так:

 90 PRINT "До свиданья!":END
Как легко понять, программа закончит свою работу при условии С=7. Мы можем улучшить работу программы, добавив строку 5 и так изменив строки 80 и 90: 

5 LET N=0

 80 PRINT "Неправильно":LET C=C+1:LET N=N+1:GOTO 20

 90 PRINT "До свиданья! Вы допустили ошибок -";N:END
Рассмотрим еще одну простую программу, которую можно назвать "Русско-немецкий словарь"

 10 PRINT "Брат 1, Дочь 2, Мать 3"

 15 PRINT "Отец 4, Сестра 5, Сын 6"

 20 PRINT "Введите номер слова"

 30 INPUT C                                                                                          (38)

 40 IF C=0 THEN 120

 50 ON C GOTO 60,70,80,90,100,110

 60 PRINT "Bruder":GOTO 20

 70 PRINT "Tochter":GOTO 20

 80 PRINT "Mutter":GOTO 20

 90 PRINT "Fater":GOTO 20

 100 PRINT "Schwester":GOTO 20

 110 PRINT "Sohn":GOTO 20

 120 PRINT "Всего хорошего!":END

Работу этой программы нет нужды разъяснять. Строка 40 служит для окончания работы.

              § 6  ПОЗНАВАТЕЛЬНЫЕ ИГРОВЫЕ ПРОГРАММЫ

Рассмотрим теперь несколько игровых программ, эти игры, однако, потребуют определенных полезных размышлений, то есть это познавательные игры. Введем в дальнейшие программы следующее упрощение - будем опускать оператор LET, то есть вместо LET K=1, LET L=0 писать - A=1, L=0. Примененная версия языка Бэйсик позволяет это делать, при этом конечно предполагается выполнение операции присваивания, а не равенства.

Итак, игра - Угадай число.

10 A=INT(RND(1)*100+1)

 20 K=1:N=100:L=0

 30 PRINT "Угадай целое число, загаданное ЭВМ"

 40 PRINT "Число находится в интервале"

 50 PRINT K,N                                                                                           (39)

 60 INPUT B:L=L+1

 70 IF B=A THEN PRINT "Угадал с "L"-го раза":GOTO 10

 80 IF B>A THEN PRINT K,B:N=B:GOTO 60

 90 PRINT B,N:K=B:GOTO 60

 100 END

Как работает эта программа. Строка 10 - ЭВМ выбирает случайное целое число из интервала 1-100. Затем идет текст на экране согласно 30,40. Строка 50 печатает числа 1-100, т.е. интервал. Строка 60 - ЭВМ ожидает ввода вашего числа и засчитывается ваша попытка (число L увеличивается на единицу). Строка 70 понятна. Как работают строки 80 и 90, лучше понять из конкретного примера. Пусть, например, машина загадала число 36 (вам неизвестное). Тогда ваш разговор с машиной может быть таким:

 RUN
 Угадай целое число, загаданное ЭВМ

 Число находится в интервале

 1       100

 ?

 60                           1-я попытка

 1       60             новый интервал согласно 80

 ?

 25                           2-я попытка

 25      60           новый интервал согласно 90

 ?

 40                       и так далее

 25      40

 ?

 36

 Угадал с 4-го раза

 ОК

После этого ЭВМ возвращается к началу программы и загадывает новое число.

Следующие три программы построены по тем же простым алгоритмам, которыми мы пользовались, когда изучали программируемый микрокалькулятор [2].

Итак, программа игры - Артиллерия.

 10 L1=1000:N=0:S=20:L2=2000

 20 INPUT A:N=N+1                                                                               (40)

 30 R=3.1416*A/180:L=L2*SIN(2*R)

 40 IF (L-L1)^2<S THEN PRINT"Попал с "N"-го раза":GOTO 60

 50 PRINT "Перелет (недолет) на "L-L1" метров":GOTO 20

 60 END
В строке 10  L1 это расстояние до цели, N будет считать число выстрелов, S -площадь поражения цели, L2 -максимальная дальность стрельбы вашей пушки. В строке 20 ЭВМ запрашивает угол А - угол возвышения (в градусах) и производится выстрел. В сроке 30 градусы переводятся в радианы и вычисляется дальность полета снаряда для данного выстрела. При попадании снаряда внутрь площади поражения (строка 40) игра заканчивается, иначе (строка 50) отмечается перелет или недолет (положительное или отрицательное значение числа (L-L1) ) и запрашивается новое значение угла А. Отметим, что в языке Бейсик углы должны быть в радианах. Условия вашей стрельбы можно изменить, изменив значения исходных данных L1,S,L2.

     Следующая программа - Танковая атака.

 10 L1=1000:L2=1000:S=50:V=10:T=9

 20 INPUT A:L1=L1-V*T                                                                          (41)

 30 R=3.1416*A/180:L=L2*SIN(2*R)

 40 IF (L-L1)^2<S THEN PRINT "Победа!":GOTO 70

 50 IF L1<0 THEN PRINT "Проиграл":GOTO 70

 60 PRINT "Промах",(L-L1),L1:GOTO 20

 70 END
Напоминаем, что в этой игре вы стреляете по движущемуся на вас танку. В строке 10 задаются первоначальное расстояние до танка, максимальная дальность стрельбы (при угле 45  градусов) , площадь поражения (м2 ) скорость движения танка (м/сек), промежуток времени между вашими выстрелами (в нашем случае мы выбрали 9 сек). В строке 20 - выбор угла А для выстрела и уменьшение расстояния до танка. Окончание игры - либо при вашем попадании - строка 40, либо поражении, танк "давит" вашу пушку - строка 50.

     Наконец, приведем программу - Посадка на Луну:

 10 M=90:H=50:V=8:PRINT M,H,V

 20 INPUT Z                                                                                      (42)

 30 A=1.62-Z*0.045:M=M-Z*0.15

 40 H=H-V-A/2:V=V+A:PRINT M,H,V

 50 IF H<0.2 THEN IF V<2 THEN PRINT "Посадка!":GOTO 90

 60 IF H<0.2 THEN IF V>=2 THEN PRINT "Авария!":GOTO 90

 70 IF M<=0 THEN M=0:Z=0:GOTO 30

 80 GOTO 20

 90 END
В строке 10 задаются исходные данные: горючее - 90 кг, высота - 50 м, скорость движения ракеты к Луне - 8 м/сек. Строка 20 - вы должны выбрать и ввести величину тяги двигателя (меняется в пределах 1-100, проценты от максимальной тяги). В строках 30,40 используются уже известные нам [1,2] простые формулы для изменения ускорения, массы горючего, высоты над поверхностью Луны и скорости ракеты через одну секунду полета при выбранной тяге Z. Строка 50 определяет условия удачной посадки ракеты на Луну (высота меньше 0,2 метра, скорость меньше 2 м/сек). Строка 60 определяет условия неудачи. Строка 70 учитывает тот случай, когда вы сожгли все горючее, плохо управляя работой двигателя, тогда, конечно, тяга равна нулю и ракета летит неуправляемо под действием только силы тяготения Луны (многократный переход с 70 на 30, пока ракета не разобьется). Строка 80 относится к случаю, когда горючее в баках еще есть и можно управлять полетом, меняя тягу (переход на 20).

Набрав эту простую программу, можно легко и быстро проверить свое умение управлять таким сложным аппаратом, как ракета возле Луны. Напоминаем [1,2], что поскольку все формулы движения ракеты используются реальные, то и наш полет максимально приближен к реальной обстановке.

§ 7.  "ГРАФИКА" БЕЗ ГРАФИЧЕСКИХ ОПЕРАТОРОВ

Для рисования графических изображений (рисунков, кривых, графиков и т.п.) обычно используют так называемые графические операторы языка Бейсик. К рассмотрению этих операторов мы перейдем несколько позже, а сейчас посмотрим, как можно "рисовать" на экране ПЭВМ, применяя уже известный нам оператор PRINT TAB.

Два примера такого рисования мы уже рассматривали в § 5 этого  пособия (программы 33 и 34 -парабола и график успеваемости учеников). Приведем еще один простой пример рисования синусоиды:

  5 FOR X=0 TO 40:PRINT"-";:NEXT X
 10 FOR  X=0 TO 15 STEP .5                                                                 (43)

 20 PRINT "I":PRINT TAB(30+15*SIN(X));"*"

 30 NEXT X
 40 END
В этом примере, кроме самой синусоиды, мы нарисовали еще и координатные оси при помощи командной строки 5 и команды PRINT"I" в начале строки 20. Конечно же, недостатком является вертикальное расположение графиков при таком способе рисования. Это неудобство исчезает, как мы уже говорили,  при переносе графиков на бумагу с помощью принтера. В этом случае оператор PRINT нужно заменить на LPRINT (при этом к компьютеру должен быть подключен принтер).

Более полно возможности такого способа рисования без графических операторов покажем на примере нескольких контролирующе-обучающих и игровых программ.

В (44) приведена часть программы, проверяющей знания ученика по астрономии. Программа содержит 8 вопросов, требуется ответить на 4 из них, номера вопросов выбираются учениками. На экране дисплея по мере прохождения программы появляются изображения созвездий. Ученик должен указать правильное имя созвездия. В командах 102-107 с помощью оператора PRINT TAB на экране дисплея формируется изображение созвездия Кассиопея.

Аналогичным образом построены изображения других созвездий в оставшихся семи вопросах, переход к этим вопросам производится с помощью так называемого переключателя ON  GOTO в строке 40 нашей программы.

 3 CLS
 5 K=0:L=0:DIM M(4)                                      

 7 PRINT TAB(20)"АСТРОНОМИЯ":PRINT                                                                     (44)

 10 PRINT TAB(10)"в программе имеется 8 вопросов":PRINT
 15 PRINT TAB(8)"вам нужно ответить на 4 вопроса"

 17 PRINT"     (варианты ответов предлагаются)"

 30 IF K=4 THEN PRINT"номера ваших вопросов-";:

 32 FOR K=1 TO 4:PRINT M(K):NEXT K:PRINT"всего ошибок-"L":GOTO 70

 34 K=K+1:PRINT

 35 PRINT TAB(20)"выберите номер вопроса (1-8)":INPUT N:M(K)=N

 40 ON N GOTO 100,200,300,400,500,600,700,800

 70 IF L=0 THEN PRINT"        отлично"

 72 IF L=1 THEN PRINT"        неплохо"

 74 IF L>=4 THEN PRINT"вам надо поучиться"

 76 IF L<4 THEN IF L>=2 THEN PRINT"        слабовато"

 80 PRINT"если хотите повторить, нажмите 1, если нет - 0"

 85 INPUT Z:IF Z<>1 THEN PRINT"      ДО СВИДАНЬЯ!":GOTO 99

 90 K=1:L=0:GOTO 35

 99 END

 100 PRINT"                 ВОПРОС  №1"

 101 PRINT:PRINT

 102 PRINT TAB(14)"+"

 103 PRINT:PRINT:PRINT:PRINT

 104 PRINT TAB(18)"+",TAB(28)"*"

 105 PRINT TAB(43)"*"

 106 PRINT:PRINT:PRINT

 107 PRINT TAB(32)"*"

 108 PRINT:PRINT

 110 PRINT"варианты - 1)Лев, 2)Орион, 3)Кассиопея, 4) Дельфин"

 120 INPUT V

 130 IF V<>3 THEN PRINT"неверно, еще раз-":L=L+1:GOTO 100

 140 PRINT"правильно":GOTO 30

 200 PRINT"                 ВОПРОС  №2"

     и так далее...

Новая команда в строке 3 CLS очищает экран от всех букв и цифр перед началом работы нашей программы, конечно так удобнее. В строке 5  K - считает вопросы, на которые вы отвечаете, L - считает ошибки, допущенные вами при ответах. Об этом и сообщается вам в конце работы при помощи строки 30. M(K) - информационный массив, в который записываются номера выбранных вопросов, согласно строке 30 они на всякий случай сообщаются вам в конце работы. Остальные строки программы  Астрономия по-видимому не требуют дополнительных разъяснений.

Полностью программа Астрономия содержится в файле astron.bas , находящемся в директории BASIC. Необходимо заметить, что программа (44) имеет недостатки - ученик должен сам выбирать номер вопроса, используется только выборочный вариант поиска правильного ответа (строка 110). Однако можно сделать так, чтобы компьютер сам произвольным образом выбирал для ученика вопросы. Для этого в программу (44) нужно добавить следующие строки:

6 RANDOMIZE(TIMER)

8 DIM A(8)

35 N=INT(1+8*RND(1)):IF A(N)=1 THEN 35                                  (45)

36 A(N)=1

37 T=T+1

38 M(K)=N

78 IF T=8 THEN 99


В строке 35 номер вопроса теперь задается случайным образом, использование переменной A(N) в конце строки и в строке 36 не позволяет снова задать тот вопрос, который уже задавался раньше. В строке 8 задается размер переменной A(N). В строке 6 используется новый оператор языка Бейсик, здесь он задает случайным образом начало работы строки 35 (в зависимости от времени запуска программы 44, определяемого по встроенному в компьютер счетчику времени). Строки 37 и 78 прекращают работу программы, когда вопросы закончатся.


Такие изменения сделаны в файле astron1.bas, (директория BASIC). Кроме того в него внесены необходимые изменения, обеспечивающие разнообразие выбора  правильного ответа - теперь иногда нужно определить правильный ответ из вариантов ответа, последовательно появляющихся на экране, иногда нужно просто самому напечатать правильный ответ. Первый случай осуществится, если к примеру в вопросе №1 строки после 108-й записать так:

110 PRINT(ЛЕВ(
112 INPUT N:IF N=1 THEN PRINT(неверно(:L=L+1

114 PRINT(ОРИОН(                                                                                          (46)

116 INPUT N:IF N=1 THEN PRINT(неправильно(:L=L+1

118 PRINT(ДЕЛЬФИН(
120 INPUT N:IF N=1 THEN PRINT(ошибка(:L=L+1

122 PRINT(КАССИОПЕЯ( 

124 INPUT N:IF N=1 THEN PRINT(верно(:GOTO 30

126 PRINT(неверно, правильный ответ - КАССИОПЕЯ(:L=L+1:GOTO 30


В 46 предполагается, что ответ - да - соответствует нажатию 1 (N=1), а ответ – нет - нажатию другой цифры (например 0).


Во втором случае строки будут выглядеть так:

120 M=0

122 INPUT N$

124 IF N$=(КАССИОПЕЯ( THEN PRINT(верно(:GOTO 30                                     (47)

126 IF M=2 THEN 130

128 PRINT(ошибка, попробуйте еще раз(:M=M+1:GOTO 122

130 PRINT(неверно, правильный ответ - КАССИОПЕЯ(:L=L+1:GOTO 30


В 47 мы используем новое понятие - строковая переменная  (N$). Теперь в ответ на запрос программы в строке 122 нужно вводить не число, а любое слово. Если это слово в нашем случае - КАССИОПЕЯ, то строка 124 определит, что ответ верен. Если ученик наберет неправильное название созвездия, то он сможет ошибиться не более двух раз (строки 120, 126, 128), после чего ему будет выдан правильный ответ и засчитана ошибка (строка 130). Заметим, что в 47 предполагается при ответе на вопрос использовать заглавные буквы, поэтому ответы : кассиопея, Кассиопея  не будут считаться правильными . Нельзя также помещать пробелы до или после ввода слова.


Итак, в программе  Астрономия1 использованы описанные три варианта предоставления возможности ответов на заданные вопросы.

Подобным образом могут быть составлены программы для проверки знаний и по другим предметам ( геометрия, география -контуры государств на карте, контуры островов, озер и т.п.), ботаника (изображения растений), черчение, зоология (изображения животных, насекомых) и так далее.

Следующая программа Скорпион (48) основана на простой физической задаче движения брошенного тела по параболе под действием силы тяжести. Первобытный охотник, стоящий на вершине скалы, должен бросить камень и попасть в скорпиона, бегающего внизу и стремящегося спрятаться под скалой. В строке 10 программы (50) величины Z1 и S1 задают размеры изображения на экране, величина N считает броски охотника и тем самым отмеряет время с начала "охоты". Здесь же используется новый для нас оператор языка Бейсик GOSUB (переход к подпрограмме). В данном случае подпрограмма (строки 190,200) "рисует" на экране изображение охотника. Команда RETURN (ВОЗВРАТ) в строке 200 возвращает работу программы к очередной строке 20. Строки 20-40 изображают при помощи простой формулы "скалу". Строка 50 обращается к подпрограмме рисования изображения "скорпиона". Команды 60,70 осуществляют диалог с учеником с целью задания исходных данных задачи (выбор скорости скорпиона и силы броска охотника). После этого на экране формируется следующая часть кадра - команды 100-130 рисуют траекторию камня и одновременно профиль обрыва скалы, команды 140 и 210 рисуют скорпиона на новом расстоянии от края экрана S, определяемом выбранной скоростью скорпиона и величиной N. Траектория камня - парабола (формулу в строке 110 легко представить в виде    Z+1=Const (X-L1)2 ).

 10 CLS:Z1=6:S1=40:N=1:GOSUB 190

 20 FOR Z=0 TO Z1

 30 L=6-6*SQR(Z/Z1):PRINT TAB(L)"Щ" GOTO 30

 40 NEXT Z

 50 S=S1:GOSUB 210

 60 PRINT"скорость? (1-2-3)":INPUT V                                   (48)

 70 PRINT"сила броска? (0-100)(:INPUT F
 80 GOSUB 190

 90 L1=7:X1=L1+.5*F*SQR((Z1+2)/Z1)

 100 FOR Z=0 TO Z1

 110 X=L1+.5*F*SQR((Z+1)/Z1):L=6-6*SQR(Z/Z1)

 120 PRINT TAB(L)"Щ";TAB(X)"о"

 130 NEXT Z

 140 S=S1-N*V*5:GOSUB 210

 150 IF ABS(X1-S-2)<1.5 THEN PRINT"Попал!":GOTO 180

 160 IF S<5 THEN PRINT"Неудача!":GOTO 180

 170 N=N+1:PRINT"Промах!":GOTO 70

 180 END

 190 PRINT TAB(5)"*":PRINT TAB(4)"<@+-"

 200 PRINT TAB(4)"/> о":RETURN

 210 PRINT TAB(S)">00=-,":RETURN
Величина X1 это дальность полета камня. Команда 150 в случае совпадения с точностью до 2 величин X1 и местоположения головы скорпиона (S+2) фиксирует попадание. Команда 160 описывает тот случай, когда скорпион успел убежать под скалу. Команда 170 фиксирует промах, увеличивает N на единицу (это соответствует последующему перемещению скорпиона по экрану влево) и запрашивает новую силу броска (70). И так далее..

Первый "кадр" выполнения задачи Скорпион изображен на Рис. 2. Мы выбрали скорость 2 и силу броска 70. После этого на экране появляется следующий кадр (Рис. 3) нашего "диафильма", причем скорпион перемещается на расстояние 5V влево, а на экране появится траектория движения камня. Траектория появляется точка за точкой по мере перемещения всего кадра вверх по экрану, образуется своеобразная иллюзия падения брошенного камня.

                                                                           Рис. 2

По мере развития событий кадры будут меняться до тех пор, пока не произойдет попадание в цель или ускользание скорпиона под скалу.

Программы подобного типа могут быть легко составлены для решения похожих задач на уроках физики и астрономии.

                                                                           Рис. 3

Например, можно учесть силу попутного или встречного ветра, учесть влияние сопротивления воздуха. Можно подобным образом составить программу по изучению понятия первой космической скорости -горизонтальная стрельба из пушки, стоящей на вершине высокой горы  (известная задача Ньютона). В этом случае надо принять во внимание кривизну земной поверхности и использовать закон тяготения Ньютона в его общем виде.

Рассмотрим теперь программу Космодром (49). Эту программу мы уже видели в задачах "Посадка на Луну" для микрокалькуляторов МКШ-2 и МК-56 [1,2] ,для ПЭВМ Д3-28 и ДВК-1 [ 4] и для учебного класса IBM (§ 6). По сравнению с упрощенным вариантом, рассмотренным в (§ 6), ниже приведенная программа усилена изображениями звездного неба и летящей ракеты, таким образом  диалог с ПЭВМ иллюстрируется "диафильмом", рисующим на экране отдельные кадры полета ракеты к поверхности Луны (Рис.4).

 10 CLS:V=8:M=90:H=30

 20 PRINT TAB(53)"*":PRINT:PRINT:PRINT:PRINT

 30 PRINT TAB(12)"*";TAB(22)"*";TAB(34)"+";TAB(56)"*"

 40 PRINT:PRINT"+":PRINT TAB(41)"*"

 50 R=10:Y=6:L=INT(H*Y/30)

 60 IF H<1 THEN IF V>2 THEN  150

 70 IF H<1 THEN IF V<=2 THEN  160                                                       (49)

 80 PRINT TAB(R)"-oo-"

 90 FOR K=1 TO L:PRINT"I":NEXT K

 100 PRINT “ПППППППППППППППП”

 110 PRINT M,H,V

 120 IF M<1 THEN M=0:Z=0:GOTO 140

 130 PRINT"тяга? (0-100)":INPUT Z

 140 A=1.62-Z* .045:M=M-Z* .15:H=H-V-A/2:V=V+A:N=N+1:GOTO 20

 150 PRINT TAB(R+1)"\!/":PRINT TAB(R)"@-*-@":END

 160 PRINT TAB(R)"-oo-":PRINT TAB(R+1)"MM":END

Сравнивая программу (49) с программой Посадка на Луну из  (§ 6), видим что главные отличия заключаются в наличии строк 20-40 -это изображение звездного неба, строк 80-90 -это рисунок нашей ракеты и "пустого" пространства высотой L под нею. Изменены также строки, описывающие окончание полета.

Ниже приведен первый кадр (Рис. 4) полета ракеты к Луне.

                                                                                         Рис. 4

Согласно строкам 150,160 программы (49) возможны два варианта окончания - корабль разбился и корабль совершил мягкую посадку:

                                                              Рис. 5

Как мы видели в работе [ 4], выполнение программы Космодром выглядит более зрелищно, чем только что рассмотренное. Это связано с тем, что мы использовали только 16 строк на экране дисплея (это дает возможность использования программы для ПЭВМ типа Корвет). А для используемых в  [ 4] ПЭВМ на экране размещается 25 строк. Это позволяет добавить изображения летающей тарелки и лунохода, двигающихся независимым от ученика образом, нарисовать "выхлоп" реактивной струи из двигателя нашей ракеты, причем размер выхлопа зависит от выбранной тяги, и так далее. Такие же замечания относятся и к программе Скорпион - в [ 4] изображение скорпиона более естественное (оно занимает три строки на экране дисплея). 

В заключение приведем использованные в строке 140 формулы  программы (49) в более привычном  виде:

 m = m -0,15 Z t  - изменение массы горючего при времени работы двигателя t с силой тяги Z. Коэффициент 0,15 задан (он связан с величиной скорости истечения газов из сопла ракеты)

a =

- 

 - величина ускорения ракеты при одновременном действии силы притяжения Луны (в шесть раз слабее, чем на Земле) и силы тяги двигателя F - максимально возможное значение силы тяги (36000 Н в нашем случае), M - масса ракеты (8000 кг)

h = h - v t - 

     ,   v = v + a t       эти две формулы не требуют, надеемся, разъяснений.


Программы  Скорпион  и  Космодром можно сделать более выразительными, пойдя на некоторые усложнения. Это сделано в файлах  scorp.bas и kosmo.bas, имеющихся в директории BASIC. Так в программе Космодром  добавлены изображения летающей тарелки и лунохода, двигающихся независимым от ученика образом, нарисованы "выхлоп" реактивной струи из двигателя нашей ракеты, причем размер выхлопа зависит от выбранной тяги. На Рис. 6 показан один из кадров работы такой программы.

Рис. 6     

                  § 8. ГРАФИЧЕСКИЕ ОПЕРАТОРЫ

     Графические операторы языка Бейсик позволяют, уже по настоящему, рисовать на экране ПЭВМ. В таком случае экран можно представить в виде планшета с осями координат - горизонтальной и вертикальной. Для ПЭВМ IBM 286 по горизонтали может разместиться 500 (в одном из графических режимов) отдельных точек графического изображения, по вертикали - 250 точек; направление горизонтальной оси обычное - слева направо, а вот вертикальная ось направлена сверху вниз. Однако для экрана ПЭВМ такое направление кажется естественным, так как строчки на экране также появляются сверху. Таким образом начало координат расположено в левом верхнем углу экрана.

     Рассмотрим простейшую графическую программу:

 10 CLS
 20 INPUT X,Y
 30 PSET (X,Y)                                                             (50)

 40 GOTO 20

     В строке 30 PSET X,Y означает - ПОСТАВЬ ТОЧКУ с координатами X и Y. Величины координат запрашиваются в строке 20, мы уже знаем, что X не должно превышать 500, а Y - 250. Программа (56) позволит создать на экране "звездное небо", постепенно изображая точки в разных местах экрана.

     Следующая программа позволит стереть с экрана по очереди все ранее нарисованные точки:

 10 CLS

 20 INPUT X,Y

 30 PRESET (X,Y)                                                       (51)

 40 GOTO 20

     Таким образом, оператор PRESET означает - СОТРИ ТОЧКУ. Напоминаем, что для остановки работы программ (56) и (57) нужно применить красную клавишу СТОП.

     Перейдем теперь к рисованию прямых линий. Это и делает программа :

 10 CLS
 20 INPUT X,Y
 30 LINE (0,0)-(X,Y)                                                    (52)

 40 GOTO 20

     Итак LINE (0,0)-(X,Y) означает - НАРИСУЙ ЛИНИЮ между точками с координатами 0,0 и X,Y.     Если мы в конце строки 30 программы (52) добавим букву B, то сможем нарисовать вместо линии прямоугольник, вершинами которого являются точки (0,0) и (X,Y), расположенные по его диагонали:

 10 CLS

 20 INPUT X,Y

 30 LINE (0,0)-(X,Y),7,B                                                        (53)

 40 GOTO 20

Заметим, что одновременно добавленная в строку цифра 7 означает, что рисунок будет белого цвета. Можно выбрать и другой цвет, если вместо 7 подставлять цифры в интервале 1-15. 

     Внесем еще одно изменение в нашу программу - добавим в конце строки 30 вместо буквы B сразу две буквы - BF. В этом случае программа будет рисовать закрашенные прямоугольники. После многократного исполнения программ (52) и (53) экран окажется заполнен изображениями линий и прямоугольников. Для стирания с экрана уже ненужных графических изображений служит  команда CLS. Наберите ее, нажмите клавишу ENTER и экран станет чистым.

     Перейдем к рисованию окружностей. Следующая программа нарисует окружности радиусом 40 с центрами в точках, координаты которых вы будете указывать по мере выполнения программы.

 10 CLS

 20 INPUT X,Y

 30 CIRCLE(X,Y),40                          (54)

 40 GOTO 20

     Итак, оператор CIRCLE(X,Y),40 рисует окружность радиусом 40 с центром в точке с координатами X,Y. Следующая программа (55) нарисует в центре экрана ряд концентрических окружностей различного радиуса.

 10 CLS

 20 INPUT R

 30 CIRCLE(250,125),R                                      (55)

 40 GOTO 20

     А теперь немного отдохнем, рисуя на экране замысловатые узоры с помощью программы (56).

 10 CLS

 20 INPUT A,B

 30 FOR Z=0 TO 6.28 STEP .01

 40 R=SIN(5*Z)                                                       (56)

 50 X=250+120*COS(A*Z)*R

 60 Y=120+120*SIN(B*Z)*R

 70 PSET(X,Y)

 80 NEXT Z
     Как следует из текста программы, узоры образуются оператором PSET, координаты точек создаются при помощи цикла формулами 40- 60, причем константы A и B, определяющие форму узора, задаются вами командой 20. A и B должны быть целыми числами в интервале 1-10, при больших значениях изображения менее интересны.  Два из возможных узоров изображены на рисунках 7 и 8.

                                                                                   Рис. 7

                                                                                       Рис. 8      

        Следующая программа (57) рисует ряд разноцветных синусоид, цвет задается величиной С в строке 60, амплитуда - величиной В. Строки 30,40 рисуют координатные оси.

 10 CLS

 20 INPUT B,C

 30 LINE(0,120)-(500,120)

 40 LINE(250,0)-(250,240)                         (57)

 50 FOR X=0 TO 500

 60 PSET(X,120+B*SIN(X*6.28/500)),C

 70 NEXT X
     Имеется также возможность закрашивания фигур, образованных на экране при выполнении какой-либо программы. Так, программа (58) рисует ряд концентрических кругов, закрашенных разными цветами. Оператор PAINT(250,120),I закрашивает в нашем случае круг с центром в точке (250,120) цветом I. В соответствии со строкой 30, круги создаются окружностями с меняющимися радиусом (от 56 до 8) и цветом (от 7 до 1, то есть все цвета радуги).

 10 CLS

 20 FOR I=7 TO 1 STEP -1

 30 CIRCLE(250,120),I*8,I                          (58)

 40 PAINT(250,120),I

 50 NEXT I

     Увеличим теперь возможности программы (58), добавим строки 15 и 60, изменим строку 30:

 10 CLS

 15 INPUT K

 20 FOR I=7 TO 1 STEP -1

 30 CIRCLE(250,120),I*8,I,,,K                       (59)

 40 PAINT(250,120),I

 50 NEXT I
 60 GOTO 15

     В конце строки 30 мы добавили три запятые и букву K. Вводя при помощи строки 15 значения величины K, мы получим вместо окружностей эллипсы, сплюснутые по вертикали или горизонтали в зависимости от того, больше или меньше единицы величина K. Лишние запятые в нашем случае не дают ничего, однако, если бы мы захотели построить с помощью оператора CIRCLE не замкнутые окружности (эллипсы), а их части (дуги), можно было бы поместить между запятыми значения начала и конца дуги, выраженные в радианах (от -6,28 до +6,28). Строка 60 позволит нам совместить на экране изображения нескольких эллипсов и получить эффектную картинку,  подобную представленной на рисунке 9.

                                                                              Рис. 9

     Имеется еще один оператор языка Бэйсик -COLOR, с помощью которого можно задавать цвет всего экрана. Работу этого оператора демонстрирует следующая программа:

 10 CLS
 20 N=18:DIM X(N),Y(N)

 30 R=80:T=0:D=2*3.1416/N

 40 FOR I=1 TO N

 50 T=T+D

 60 X(I)=240+R*COS(T)

 70 Y(I)=120-R*SIN(T)

 80 NEXT I

 90 FOR C=2 TO 7                                                           (60)

 100 CLS

 110 COLOR C,C-2

 120 FOR I=1 TO N-1

 130 FOR J=I+1 TO N

 140 LINE (X(I),Y(I))-(X(J),Y(J))

 150 NEXT J

 160 NEXT I

 170 NEXT C

     В строке 20 программы (60) задаются два массива координат точек, между которыми впоследствии будут проводиться прямые линии , образующие цветные узоры. В цикле 40-80 задаются и считываются координаты точек. В трех циклах, вложенных друг в друга 90-170, меняется цвет экрана и линий и образуется сам узор из переплетенных линий.

      Приведем теперь несколько простых и тем не менее очень эффектных программ, позволяющих создавать на экране ПЭВМ красивые ,так называемые "муаровые" узоры. Программа (70)просто рисует ряд прямых линий между точками, координаты точек заданы при помощи цикла 30-40.

 10 CLS

 20 FOR Y=15 TO 220 STEP 2

 30 LINE(15,15)-(400,Y)                                                   (61)

 40 NEXT Y
Результат работы программы приведен на рисунке 10.

                                                                                      Рис. 10

    Следующая программа (62) использует для создания узора рисование большого количества налагающихся друг на друга окружностей.

 10 CLS

 20 FOR R=1 TO 120 STEP 2

 30 CIRCLE(200,125),R                             (62)

 40 CIRCLE(270,125),R

 50 NEXT R

     Получающийся узор приведен на рисунке 11. Заметим, что можно поменять в этих программах шаг цикла (в небольших пределах 1-3) и наблюдать, как при этом изменяется узор.

     Еще одна программа (63) использует полярные координаты (в строке 20 меняется угол от нуля до 90 градусов) для рисования веера радиусов (строки 40,50). В строке 60 рисуем дополнительный веер, соединяя концы радиусов с точкой (90,60)

                                                                                 Рис. 11

      Обратим внимание, что в этой строке используется еще одна форма написания оператора LINE, в этом случае линия будет проводиться между точкой, указанной в предыдущей строке (X2,Y2) и точкой (90,60). Получающийся узор приведен на рисунке  12. 

 10 CLS: X1=30:Y1=30:R=220

 20 FOR F=0 TO 90

 30 A=F*3.1416/180                                                                         (63)

 40 X2=X1+R*SIN(A):Y2=Y1+R*COS(A)

 50 LINE(X1,Y1)-(X2,Y2)

 60 LINE -(90,60)

 70 NEXT F

                                [image: image13.png]TN

HTAREROS
,mm‘\'l. IS

fi
nm«f« 1N

AN
i





                                                                                        Рис. 12

     В заключение параграфа приведем еще две программы  несколько более сложные, но приводящие к эффектным рисункам на экране ПЭВМ. Программы используют оператор LINE и цветовые возможности оператора PAINT. Итак, первая программа, позволяющая рисовать что-то  вроде переливающихся цветами радуги многогранников:

 10 CLS

 20 X1=250:Y1=125

 30 INPUT A,B

 40 FOR I=0 TO 360 STEP 4

 50 Z2=3.1416*I/180

 60 Z3=3.1416*(I+4)/180                                                                               (64)

 70 X2=250+70*SIN(5*Z2)*COS(A*Z2)

 80 X3=250+70*SIN(5*Z3)*COS(A*Z3)

 90 Y2=125+70*COS(5*Z2)*SIN(B*Z2)

 100 Y3=125+70*COS(5*Z3)*SIN(B*Z3)

 110 C=RND(1)*5+1

 120 LINE(X1,Y1)-(X3,Y3),C

 130 X1=X2:Y1=Y2

 140 NEXT I

 150 END

     В строках 20 и 30 задаются координаты центра картины и ПЭВМ просит ввести два любые числа (целые, лучше в интервале 1-5). Строка 40 - начало цикла рисования многогранника, I -угол в градусах, которые в строках 50,60 переводятся в радианы для двух очередных вершин многогранника. В строках 70-100  по приведенным формулам вычисляются координаты  этих вершин, строка 120 рисует прямую, соединяющую точки (X1,Y1) и (X3,Y3). Таким образом, точка (X2,Y2) пропускается, но строка 130 использует эту точку в качестве начальной при рисовании следующей грани многогранника, и так  далее, в конечном счете на экране образуется как бы сдвоенный многогранник, грани которого образуются соединением отдельно соседних четных и соседних нечетных его вершин. Цвет каждой грани задается случайным образом при помощи строки 110.

     На рисунке 13 представлено одно из возможных изображений, получающихся в результате исполнения программы. Кроме выбора значений А и В для получения разнообразных вариантов образующейся фигуры, можно заменить число 4 в строках 40 и 60 на другое число (2,3,5,6...), то есть такое число, чтобы на 360 градусов укладывалось целое число граней. Но на наш взгляд число 4 приводит к более эффектному рисунку.

     Следующая программа  (65)  , используя те же формулы, рисует снова многогранники, но уже не сдвоенные, однако с поочередно закрашиваемыми всеми цветами радуги плоскими гранями, так что на экране образуется нечто вроде мерцающего разноцветного фонарика.

                                                                         Рис. 13

 10 CLS

 20 X1=250:Y1=125

 30 INPUT A,B

 40 FOR I=0 TO 360 STEP 4

 50 Z=3.1416*I/180

 60 X=250+90*SIN(5*Z)*COS(A*Z)

 70 Y=125+90*COS(5*Z)*SIN(B*Z)

 80 LINE(X1,Y1)-(X,Y),7

 90 X1=X:Y1=Y

 100 NEXT I                                                                                                    (65)

 110 FOR R=1 TO 70 STEP 3

 120 FOR A=0 TO 6.28 STEP .8

 130 U=250+R*COS(A)

 140 V=125+R*SIN(A)

 150 C=RND(1)*5+1

 160 PAINT(U,V),C,7

 170 NEXT A:NEXT R
 180 END
     Строки 10-100 по-видимому не требуют объяснения, в двойном цикле 110-180 при помощи полярных координат задается вращающийся вокруг центра отрезок, длина которого постепенно увеличивается. Конец отрезка (U,V) должен попадать поочередно внутрь каждой плоской грани нарисованного многогранника, и в соответствии со строками 150,160 каждая грань будет поочередно закрашиваться каким-то случайно выбранным цветом.

     Цветное изображение одного из таких фонариков представлено на рисунке 14 . При выполнении этой программы иногда может получиться так, что конец отрезка (U,V) выйдет за пределы нарисованного фонарика, тогда закрашиваться начнет уже все свободное пространство экрана, общая картина получается при этом тоже интересной, но закраска экрана происходит слишком медленно. В этом случае для прерывания работы программы можно использовать нажатие клавиш Ctrl Pause  в один из моментов, когда экран полностью закрашен каким либо цветом.

                                                                    Рис. 14

 § 9 РИСОВАНИЕ "МУЛЬТФИЛЬМОВ" 

     Простое перемещение по экрану движущегося предмета можно осуществить с помощью программы (66).

 10 CLS

 20 FOR X=10 TO 200 STEP 5

 30 CIRCLE (X,X),5,7                                             (66)

 40 CIRCLE (X,X),5,0

 50 NEXT X

     В строке 30 рисуется кружок радиусом 5 и цветом 7 (белый), а в строке 40 этот кружок рисуется цветом 0 (черный!), то есть он сразу же и стирается. Последовательные значения новых координат центра кружка задаются циклом 20-50. В результате выполнения программы создается иллюзия движения кружка из левого верхнего угла экрана вниз по диагонали.

     Изобразим на экране движение качающегося маятника часов. Для этого служит программа (67).

 10 CLS
 20 X1=120:Y1=10:R=150

 30 B=-60:C=60:D=5

 40 GOSUB 100                                                                 (67)

 50 B=60:C=-60:D=-5

 60 GOSUB 100

 70 GOTO 30

 80 END

 100 FOR F=B TO C STEP D

 110 A=F*3.1416/180

 120 X2=X1+R*SIN(A):Y2=Y1+R*COS(A)

 130 LINE(X1,Y1)-(X2,Y2)

 140 CIRCLE(X2,Y2),5

 150 CLS
 160 NEXT F
 170 RETURN
      Последовательные положения маятника нашего мультфильма изображаются с помощью под- программы 100-170, команда 150 стирает целиком с экрана предыдущее изображение маятника. Команда 30 задает движение маятника по экрану слева направо, а команда 50  -  справа налево. Следующая программа (68) создает вращение прямой линии против часовой стрелки и одновременное круговое движение по часовой стрелке (навстречу линии) небольшого круга.

 10 CLS

 20 X1=170:Y1=90:R=50

 30 FOR F=0 TO 360 STEP 5

 40 A=F*3.1416/180

 50 X2=X1+R*SIN(A):Y2=Y1+R*COS(A)                                      (68)

 60 X3=X1-R*SIN(A)

 70 LINE(X1,Y1)-(X2,Y2)

 80 CIRCLE(X3,Y2),10

 90 PCLS

 100 NEXT F

 110 GOTO 30

     В программах (67) и (68) в отличие от (66) сменяющиеся кадры "мультфильма" создаются с помощью оператора CLS, полностью стирающего очередное изображение с экрана. Ясно, что такой способ не может быть применен, когда во время смены кадра изменяются только некоторые фигуры экрана, а другие остаются неизменными. Это замечание как раз относится к следующей программе (69), создающей на экране мультфильм о совместном движении Земли и Луны вокруг Солнца. В программе (69), как и в (66), применяется стирание очередного изображения заменой его цвета на черный (C=0).

 10 CLS
 20 CIRCLE(240,120),15,7                                                изображение

 30 PAINT(240,120),4,7                                                    Солнца

 40 F2=30

 50 FOR F1=5 TO 360 STEP 5

 60 A1=F1*3.1416/180:A2=F2*3.1416/180

 70 C=0:K=0

 80 GOSUB 300

 90 X1=240+70*COS(A1)                                                   координаты
 100 Y1=120+70*SIN(A1)                                                   Земли

 110 C=7:K=2                                                                       цвет Земли

 120 GOSUB 300

 130 C=0:L=0                                                                                                         (69)

 140 GOSUB 400

 150 X2=X1+15*COS(A2)                                                    координаты

 160 Y2=Y1+15*SIN(A2)                                                     Луны

 170 C=7:L=3                                                                         цвет Луны

 180 GOSUB 400

 190 F2=F2+30

 200 NEXT F1

 210 GOTO 40

 220 END
 300 CIRCLE(X1,Y1),5,C                                                       изображение

 310 PAINT(X1,Y1),K,C                                                         Земли
 320 RETURN

 400 CIRCLE(X2,Y2),3,C                                                       изображение
 410 PAINT(X2,Y2),L,C                                                         Луны
 420 RETURN
     Применим этот же прием в следующей программе (70), которая создает на экране компьютера "мультфильм" с изображением бегущего человечка.

 10 CLS

 20 FOR X=0 TO 400 STEP 20

 30 C=7

 40 GOSUB 200

 50 C=0

 60 GOSUB 200

 70 C=7                                                                                             (70)

 80 GOSUB 300

 90 C=0

 100 GOSUB 300

 110 NEXT X 

 120 END

 200 LINE(X+14,150)-(X+20,133),C

 210 LINE -(X+15,120),C

 220 LINE -(X+20,93),C

 230 LINE(X+5,145)-(X+25,133),C

 240 LINE -(X+15,120),C

 250 LINE(X+20,115)-(X+10,110),C

 260 LINE -(X+19,92),C

 270 LINE -(X+20,108),C

 280 LINE -(X+30,113),C

 290 LINE(X+20,83)-(X+26,89),C,B

 295 RETURN

 300 LINE(X+2,132)-(X+25,136),C

 310 LINE -(X+40,93),C

 320 LINE(X+43,150)-(X+50,130),C

 330 LINE -(X+30,120),C

 340 LINE(X+30,111)-(X+22,103),C

 350 LINE -(X+38,95),C

 360 LINE -(X+43,110),C

 370 LINE -(X+58,104),C

 380 LINE(X+40,83)-(X+46,89),C,B

 390 RETURN

     Подпрограммы (200) и (300) изображают два последовательных кадра - фигуры бегуна в разных позах, основная же программа при помощи цикла располагает эти кадры на экране слева направо, сразу же их уничтожая и создавая тем самым иллюзию движения.

     На рисунке 15 представлено получающееся изображение (но без последовательного уничтожения кадров).

                                                                        Рис. 15
     Рассмотрим теперь еще раз программу Танковая атака, рассмотренную нами ранее в  § 6. Но теперь мы используем графические операторы, что позволит нам существенно "оживить" программу:

 10 CLS

 20 PI=3.1416:G=9.8:S=450:V=70

 30 GOSUB 300

 40 C=7:GOSUB 400

 50 INPUT A

 60 LOCATE 1,1

 70 VX=V*COS(A*PI/180)

 80 VY=V*SIN(A*PI/180)                                                          (71)

 90 T1=2*VY/G:X1=VX*T1

 100 FOR T=0 TO T1 STEP .1

 110 X=VX*T

 120 Y=VY*T-(G*T^2/2)

 130 Z=200-Y

 140 CIRCLE(X,Z),2,7

 150 CIRCLE(X,Z),2,0

 160 NEXT T

 170 L=15:GOSUB 500

 180 C=0:GOSUB 400

 190 IF (X1-S)^2<60 THEN PRINT "УРА!":L=40:GOSUB 500:END

 200 IF S<50 THEN PRINT "SOS!":GOSUB 400:END

 210 S=S-30:C=7:GOSUB 400

 220 GOSUB 300

 230 GOTO 50

 300 LINE(0,200)-(15,196)

 310 CIRCLE(15,196),4

 320 LINE -(10,190)

 330 LINE(6,195)-(20,190)

 340 LINE(0,200)-(500,200)

 350 RETURN

 400 CIRCLE(S,190),9,C,,,.2

 410 CIRCLE(S,195),15,C,,,.2

 420 LINE(S-20,190)-(S-10,190),C

 430 RETURN

 500 FOR K=1 TO 20

 510 F=INT(RND(1)*180)

 520 R=INT(RND(1)*L)

 530 A=X1+R*COS(F*PI/180)

 540 B=200-R*SIN(F*PI/180)

 550 LINE(X1,200)-(A,B)

 560 NEXT K

 570 RETURN

     В строке 20 нашей программы  PI  это число , G - ускорение свободного падения, S -задаем расстояние до противника в начале, V -задаем скорость полета снаряда ( в условных единицах, так, чтобы получилась подходящая по размеру картинка на экране). Строка 30 - отправляемся к подпрограмме 300, которая рисует изображение пушки и горизонтальной нижней границы кадра (строка 340). Строка40 - выбираем цвет (белый) и переходим к подпрограмме 400 (рисуем изображение танка нашего противника, находящегося в начале сражения в правой части кадра -S=450). Строка 50 - выбираем угол возвышения ствола нашей пушки при очередном выстреле. В строке 60 мы используем новый для нас оператор языка Бейсик. LOCATE 1,1 означает, что все сообщения при диалоге с ПЭВМ будут печататься в первой строке и в первой позиции строки, то есть в левом верхнем углу экрана. В строках 70-90 вычисляются по известным со школы формулам горизонтальная и вертикальная составляющие скорости снаряда при выстреле, время до падения снаряда на землю в конечной точке его траектории движения и расстояние до этой точки (дальности стрельбы при выбранном угле А).

     При помощи цикла 100-160 создается изображение летящего по параболе снаряда до его падения. Строки 110,120 используют известные формулы, строка 130 учитывает то, что для ПЭВМ координата Y отсчитывается сверху вниз, а не снизу вверх, как обычно. Строки 140,150 рисуют и сразу же стирают изображение снаряда. Строка 170 задает значение величины L и переходит к подпрограмме 500 рисования изображения разрыва снаряда на месте его падения. При помощи цикла рисуется изображения 20 отрезков, выходящих из точки падения  (X1, 200). Координаты концов отрезков (А,В) задаются при помощи полярных координат F и R , величины которых меняются случайным образом в интервале 0-180 градусов для полярного угла F и в интервале 0-L для радиуса R.

     После этого командой 180 стирается рисунок танка на старом месте, командой 210 создается рисунок танка на новом месте левее прежнего на 30 позиций (так мы выбрали), командой 220 снова рисуется пушка (это надо сделать, чтобы убрать искажения от предыдущего полета снаряда) и командой 230 мы выходим на новый выстрел и так далее...до тех пор, пока вы или попадете в танк при выполнения условия команды 190, или танк вас задавит (200).

     Один из моментов "сражения" примерно иллюстрируется рисунком (16), на самом деле пунктирная траектория полета снаряда согласно программе не будет создаваться.

                                                                                    Рис. 16                      

     Вы можете самостоятельно внести изменения в программу: замедлить или ускорить движение танка (формула S=S-30 в строке 210), облегчить или усложнить вероятность попадания в танк (формула (X1-S)^2<60) в строке 190, можно также изменить скорость полета снаряда (V в строке 20).

     Таким образом, мы с вами научились создавать достаточно интересные игровые программы, эффективно используя то, что мы уже умеем делать.

                        § 10 ИСПОЛЬЗОВАНИЕ СТРОКОВЫХ ПЕРЕМЕННЫХ

     Рассмотрим работу следующей короткой программы:

 10 CLS

 20 PRINT "Как вас зовут?"

 30 INPUT N(                                                             (72)

 40 PRINT "Привет, " N(
 50 END

     Запустим программу, на экране появится вопрос, на который вы должны сразу ответить, чтобы продолжить работу программы:

 Как вас зовут?                                                   ЭВМ задает вопрос

 ?  Петя Сидоров                                                  вы отвечаете

 Привет, Петя Сидоров                                      ЭВМ приветствует вас

 OK

     Раньше мы знали, что INPUT N означает, что ЭВМ просит вас ввести численное значение для  величины N. Теперь же мы видим, что добавление значка ( позволяет ввести в память машины в  качестве величины N( любые слова, которые в дальнейшем ЭВМ может использовать в работе     программы. Итак, N(  это так называемая строковая переменная  (один раз с нею мы уже сталкивались - в программе Астрономия). Рассмотрим следующую программу, которая может быть использована для автоматического, с помощью ЭВМ, расселения жильцов в гостинице, студентов - в общежитие и т.п.

 5 CLS

 10 DIM M((3,10)

 20 C=0

 30 FOR I=1 TO 3

 40 FOR K=1 TO 10

 50 M((I,K)="X"

 60 NEXT K

 70 NEXT I                                                                                   (73)

 80 PRINT "Здравствуйте!"

 90 PRINT "Где вы хотите жить?"

 100 INPUT "этаж";L

 105 IF L>3 THEN PRINT"этажей только 3":GOTO 100

 110 INPUT "номер";N

 115 IF N>10 THEN PRINT "номеров только 10":GOTO 110

 120 IF M((L,N)<>"X" THEN 180

 130 PRINT "свободно"

 140 INPUT "как вас зовут";M¤(L,N)

 150 PRINT " спасибо, вы поселены.":C=C+1

 155 PRINT ( Следующий!( 

 160 IF C=30 THEN 200

 170 GOTO 80

 180 PRINT "занято"

 190 GOTO 90

 200 PRINT "все уже занято"

 210 PRINT "вот список:"

 215 PRINT

 220 FOR I=1 TO 3

 230 FOR K=1 TO 10

 240 PRINT M((I,K),

 250 NEXT K

 255 PRINT

 260 NEXT I

 270 END

     В строке 10 задается общее количество имен (фамилий) жильцов нашей гостиницы, предполагается 3 этажа и 10 комнат на каждом этаже. Переменная C будет пересчитывать  жильцов. В строках 30-70 все комнаты (номера) "заселяются" все одним и тем же жильцом по имени X, дальше мы поймем, зачем это делается. В строках 80-115 у вновь прибывающего ЭВМ узнает, какой этаж и номер комнаты он выбирает. В строках 130-190 новичок сообщает свое имя, или фамилию (или и то, и другое) и ЭВМ обращается к новому прибывающему (GOTO 80). Заметим, что в строке 120 проверяется, кто живет в выбранном номере, и если это тот самый X (то есть никого из вновь прибывших нет), то дальше происходит заселение (со строки 130). Если там живет не X, значит номер уже занят (переход на 180,190) и снова возврат на строку 90 (выбор свободного номера). Кроме того, в строке 160 попутно проверяется, не заняты ли уже все номера гостиницы. Если это так, то строки 200-260 печатают полный список всех жильцов, уже живущих в гостинице. Ниже приведено начало работы программы:

 Здравствуйте!

 Где вы хотите жить?

 этаж

 ? 2

 номер

 ? 8                                                                                    (74)

 свободно

 как вас зовут

 ? Петр Сидоров

 спасибо, вы поселены.

     После этого снова возникает на экране "Здравствуйте" для нового жильца. Такая программа может использоваться при продаже билетов на самолет или в кинотеатр (в последнем случае, правда, не так уж нужны имена посетителей).

               § 11 РИСОВАНИЕ ОБЪЕМНЫХ ИЗОБРАЖЕНИЙ

     Под объемным изображением мы будем понимать такое изображение на плоском экране ПЭВМ, когда ощущается его глубина. Мы будем пытаться создать на экране примерно такую картину, которая получается при фотографировании объемных фигур в таком ракурсе, когда создается ощущение глубины картины. Удобно рассмотреть все это на примерах построения объемных изображений сложных поверхностей, задаваемых некоторыми математическими функциями двух переменных.

     Предварительно рассмотрим следующую простую программу (75), рисующую на экране объемное изображение пирамиды (рис. 17).

 10 CLS

 20 LINE(120,150)-(320,150)

 30 LINE -(400,110)

 40 LINE -(270,50)                                                                    (75)

 50 LINE -(120,150)

 60 LINE(320,150)-(270,50)

 70 END

                                                                            Рис. 17

     Как мы видим, прямоугольное основание пирамиды, так же, как ее задние боковые грани нам не видны (просто мы их не рисовали). Это и создает иллюзию объемности этой фигуры.

     Перейдем теперь к рисованию более сложных картин. Сначала приведем программу (76), рисующую прямоугольное основание (рис. 18), над которым мы дальше и будем строить объемные изображения.

 10 CLS

 20 X1=100:XC=260:YC=140:Z1=50:Y=0

 30 FOR Z=-Z1 TO Z1 STEP 6

 40 FOR X=-X1 TO X1 STEP 3

 50 U=XC+X+Z                                                                                   (76)

 60 V=250-(YC+Y+Z/2)

 70 PSET(U,V)

 80 NEXT X:NEXT Z

 90 END

                                                                  Рис. 18

      В строке 20 задаются координаты центра картины (XC,YC) и размеры основания (X1,Z1). Мы предполагаем, что координатами, лежащими в плоскости основания являются X (горизонталь - ширина) и Z (горизонталь - глубина), координата Y будет соответствовать вертикали - высоте нашей будущей объемной картины. Условие Y=0 в строке 20 как раз и означает, что сейчас мы рисуем только плоское основание картины. Само основание рисуется с помощью двойного цикла 30-80, оно как бы разрезается на заданное количество отдельных пунктирных линий, которые в совокупности образуют основание. При этом используются координаты (U,V), которые мы связываем непосредственно с горизонталью и вертикалью экрана, на котором работает оператор PSET в строке 70. Таким образом, координаты X,Y и введенную дополнительно координату Z мы оставляем для задания объемного изображения. В строке 60 учитывается, что на экране ПЭВМ, как мы знаем, координата Y отсчитывается сверху вниз -250 ( максимальный размер экрана по вертикали). Из строк 60 и 70 следует, что при переходе к рисованию очередной в цикле горизонтальной линии она смещается вправо на величину Z и вверх на величину Z/2 - это и создает, как мы видим из рисунка 93, иллюзию объемной глубины картины. В соответствии с этим размеры шагов в двойном цикле также выбраны отличающимися ровно в два раза (иначе в дальнейшем будут несколько искажаться создаваемые объемные изображения).

     Теперь построим на полученном прямоугольном основании объемную фигуру, задаваемую формулами:

          

                                                 (77)

     Как следует из этих формул, получится поверхность тела вращения, симметричного относительно вертикальной линии, проходящей через центр основания (точку X=0, Z=0). Четверка в выражении для расстояния от оси до точки на поверхности нужна для того, чтобы не нарушить создаваемой нами объемной глубины. Программа для такого случая приобретет следующий вид:

 10 CLS

 20 X1=100:XC=260:YC=140:Z1=50

 30 D=2*X1+2*Z1

 40 DIM B(D)

 50 FOR S=0 TO D:B(S)=250:NEXT S

 60 FOR Z=-Z1 TO Z1 STEP 4

 70 FOR X=-X1 TO X1 STEP 2

 80 R=SQR(X*X+4*Z*Z)                                                                      (78)

 90 Y=100*EXP(-R*R/1500)

 100 S=X1+Z1+X+Z

 110 U=XC+X+Z

 120 V=250-(YC+Y+Z/2)

 130 IF V>=B(S) THEN 160

 140 B(S)=V

 150 PSET(U,V)

 160 NEXT X:NEXT Z

 170 END

      Сравнивая (96) и (93), мы видим, что появились строки 80 и 90 - это формулы, задающие нашу объемную фигуру. Если бы отсутствовали пока нам непонятные строки 30,40,50,100,130,140, то программа (96) уже нарисовала бы нам объемную фигуру, однако с тем недостатком, что были бы видны линии поверхности, расположенные на задней стороне объемного изображения, таким образом фигура казалась бы прозрачной (см. рис. 19). Указанные строки  и служат цели удаления невидимых (скрытых) областей объемного изображения. Объясним, как они работают.

                          [image: image20.png]



                                                                               Рис. 19

      В строке 100 вводится величина S, которая при выполнении двойного цикла меняется от нуля - (X= - X1, Z= -Z1) до D (при X=X1, Z=Z1). Величина D как раз и приводится в строке  30, в строке 40 D используется в качестве размера информационного массива B(S), задаваемого в строке 50. Мы задаем значение для каждого элемента массива, равное 250, то есть максимально возможному  значению координаты V. Теперь рассмотрим, как будут работать строки 130 и 140. Пока программа будет рисовать передние (видимые) кривые линии, образующие изображение, строка 130 не будет исполняться, а строка 140 будет поочередно записывать в информационный массив все новые и новые значения координат V для каждой отдельной кривой линии, образующей нашу фигуру. Но как только при постепенном прорисовывании фигуры программа доберется до невидимых линий, начнет выполняться условие строки 130 и скрытые линии, или скрытые участки  линий уже не будут рисоваться (строка 150 будет обойдена). При этом все время нужно иметь ввиду, что координата V увеличивается при движении сверху вниз по экрану, а не наоборот. Поэтому на экране для нас скрытые участки расположены естественным образом ниже предыдущих видимых участков.

                  [image: image21.png]



                                                                             Рис. 20

     Обратим внимание, что в программе (78) мы пишем X*X вместо X^2 (аналогично для других случаев возведения во вторую степень). Это делается для того, чтобы ускорить работу программы, так как операция умножения в языке Бейсик выполняется быстрее, чем возведение в степень. Результат исполнения программы (78) приведен на рис. 20, как мы видим, действительно не видны задние, скрытые области объемной поверхности.

      Построим еще одну, более интересную фигуру, которая основана на использовании следующей формулы для Y(X,Z):

              Y =  700 Sin (R/ 12) / R                                  (79)

              Зависимость R(X,Z) - такая же, как в (77).

     В этом случае мы вынуждены несколько усложнить программу рисования такой объемной фигуры, так как на ней появятся изгибы, позволяющие видеть нижние (передние) участки изогнутой поверхности, образующей нашу фигуру. Итак, программа выглядит так:

 10 CLS

 20 X1=100:XC=260:YC=140:Z1=50

 30 D=2*X1+2*Z1

 40 DIM A(D),B(D)

 50 FOR S=0 TO D:A(S)=0:B(S)=250:NEXT S

 60 FOR Z=-Z1 TO Z1 STEP 4

 70 FOR X=-X1 TO X1 STEP 2

 80 R=SQR(X*X+4*Z*Z)

 90 Y=700*SIN(R/12)/R                                                                             (80)

 100 S=X1+Z1+X+Z

 110 U=XC+X+Z

 120 V=250-(YC+Y+Z/2)

 130 IF V>=B(S) THEN 146

 140 B(S)=V

 142 IF A(S)=0 THEN A(S)=V

 144 GOTO 150

 146 IF V<=A(S) THEN 160

 148 A(S)=V

 150 PSET(U,V)

 160 NEXT X:NEXT Z

 170 END

      Мы видим, что в строках 40 и 50 появился еще один информационный массив A(S), все элементы которого вначале равны нулю, то есть минимальному значению координаты V. Новые строки 146, 148 действуют аналогично уже известным нам строкам 130, 140, но по отношению к величине A(S). Таким образом, теперь для каждого значения X запоминаются и записываются в массивы максимальное и минимальное значения Y для ранее нарисованных кривых. Все последующие отрезки кривых, попадающие в этот интервал, не рисуются. Строки 142, 144 выполняют то же, что и строки 146, 148, но работают только в начальные моменты рисования объемной фигуры.

                                                                               Рис. 21     

     На рисунке 21 приводится результат выполнения программы (80), мы видим, что на фигуре, заданной уравнением (79), видны уже и участки нижней поверхности.

     Более сильно этот эффект проявляется, если использовать в качестве исходной формулы вместо (79) следующую формулу:

              Y=30 Cos (R/8)                            (81)

Соответственно строку 90 программы (80) нужно заменить на следующую:

90 Y=30*COS(R/8)

Результат работы программы приведен на рисунке 22.

                                                                             Рис. 22

    В заключение произведем еще одну замену в программе (80) - вместо строки 90 напишем такую строку:

90 Y=30*COS(X*Z/500)                                   

и получим эффектную картину, приведенную на рисунке 23.

                                                                                  Рис. 23

     Ясно, что можно теперь использовать и другие исходные формулы в строке 90 нашей последней программы и получать разнообразные объемные изображения.

       А теперь вернемся к способу удаления невидимых частей объемного изображения, который мы использовали при рисовании пирамиды (рис. 17): просто с самого начала учтем, какие  части  фигуры  не надо рисовать. Ясно, что в таком случае мы будем вынуждены ограничиться рисованием только простых изображений, но и программа будет проще, как это и было в случае пирамиды. Итак, программа (82) и получающийся при ее выполнении рисунок 24:

 10 CLS

 20 X1=240:Y1=80:R=50:K=2

 30 B=55:C=410:D=3:E=1.5:R1=15

 40 GOSUB 100

 50 X1=270:Y1=120:K=1

 60 B=-60:C=263:D=4:E=.3:R1=9                                           (82)

 70 GOSUB 100

 80 END

 100 FOR F=B TO C STEP D

 110 A=F*3.1416/180

 120 X2=X1+K*R*SIN(A)

 130 Y2=Y1+R*COS(A)

 140 CIRCLE(X2,Y2),R1,7,,,E

 150 NEXT F

 160 RETURN

     Основная часть программы содержится в подпрограмме 100-160, строки 20,30 и 50,60 просто задают исходные значения параметров, необходимых для работы. Программа рисует два сцепленных тора, образованных при помощи большого количества сплющенных (E)  окружностей. Центры этих окружностей задаются строками 120,130 и располагаются вдоль незамкнутых эллипсов (параметры K,B,C). Величины B и C определяют начала и концы эллипсов, подобранные так, чтобы исключить рисование участков торов, которые не должны быть видны.

                                                                       Рис. 24                                                                  

               § 12 СЛОЖНОЕ ДВИЖЕНИЕ  ИЗОБРАЖЕНИЙ

     Следующая программа (83) создает вращение объемной фигуры - треугольной пирамиды.

 10 CLS

 20 X1=170:Y1=90:R=50

 25 PI=3.1416:Y0=40:K=.2

 30 FOR F=0 TO 355 STEP 5

 40 A=F*PI/180                                                                                                  (83)

 50 X2=X1+R*SIN(A):Y2=Y1+K*R*COS(A)

 60 X3=X1+R*SIN(A+2*PI/3):Y3=Y1+K*R*COS(A+2*PI/3)

 70 X4=X1+R*SIN(A+4*PI/3):Y4=Y1+K*R*COS(A+4*PI/3)

 80 C=15:GOSUB 100

 85 C=0:GOSUB 100

 90 NEXT F

 95 END

 100 LINE(X1,Y0)-(X2,Y2),C

 110 LINE(X1,Y0)-(X3,Y3),C

 120 LINE(X1,Y0)-(X4,Y4),C

 130 LINE - (X4,Y4),C

 140 LINE - (X2,Y2),C

 150 RETURN

      Для объяснения работы этой программы обратимся сначала к программе (68) из § 9, которая описывает вращение радиуса  R=50 с центром в точке X1=170, Y1=90. Как мы видим, начальные строки (10-50) программ (68) и (83) практически совпадают, величина K, меньшая единицы, приводит к тому, что точка конца радиуса (X2, Y2) вращается не по окружности, а по эллипсу. По этому же эллипсу вращаются точки (X3, Y3) и (X4, Y4), сдвинутые на 120 и 240 градусов, таким образом они образуют вершины треугольника, лежащего в основании пирамиды. Эллиптичность приводит к кажущейся объемности построения, мы увидим пирамиду как бы сбоку сверху. Сама треугольная пирамида рисуется подпрограммой 100-160, строки 100-120 соединяют вершину пирамиды с углами основания, основание рисуется строками 120-140. Эффект вращения этой объемной фигуры достигается за счет поочередного рисования (С=15) и стирания (С=0) пирамиды каждый раз в новом ее положении по мере изменения угла F. Для того, чтобы вращение не прервалось после одного оборота, нужно заменить END в строке 95 на GOTO 30.

      Рассмотрим теперь те случаи, когда требуется создать движение каких то фигур на фоне уже имеющихся на экране изображений. Сначала простая программа (84), создающая движущуюся окружность на фоне другой, неподвижной окружности.

 10 CLS

 20 CIRCLE(140,140),45,15

 100 FOR  F=0 TO 359 STEP 1

 110 A=F*3.1416

 120 X=220+60*SIN(A):Y=220+60*COS(A)                                                               (84)

 130 CIRCLE(X,Y),45,15

 140 CIRCLE(X,Y),45,0

 150 NEXT

 160 GOTO 100

      При запуске этой программы, как мы увидим, движущаяся окружность, проходя по области, занятой неподвижной окружностью, стирает все на своем пути, искажая таким образом первоначальную картину. Как избежать этого недостатка?

       С этой целью используем новые для нас операторы языка Бейсик  - GET (БРАТЬ) и PUT (ПОМЕЩАТЬ). Оператор GET запоминает в массиве изображение заданной вами прямоугольной области экрана, а оператор PUT помещает сохраненное таким образом изображение в любое заданное вами новое место. Для задания области служит запись GET(X1,Y1)-(X2,Y2),K - здесь X1,Y1 - координаты левого нижнего угла прямоугольной области, X2,Y2 - координаты правого верхнего угла, K - имя массива, в котором будет записано изображение области. Размер массива определяется значением, которое мы не будем объяснять, N= INT(((X2-X1)*2+7)/8)*(Y1-Y2). Напомним только, что вертикальная координата отсчитывается сверху вниз. В соответствии с этим дополним программу (84) строчками 60, 70, 80 и заменим строчки 130 и 140:

 10 CLS

 20 CIRCLE(140,140),45,15

 60 N=INT((90*2+7)/8)*90

 70 DIM KR(N)

 80 GET(95,185)-(185,95),KR 

 100 FOR  F=0 TO 359 STEP 1

 110 A=F*3.1416

 120 X=220+60*SIN(A):Y=220+60*COS(A)                                                               (85)

 130 PUT(X,Y),KR

 140 PUT(X,Y),KR

 150 NEXT

 160 GOTO 100

       Массив мы обозначили именем KR, в нем записана прямоугольная область размерами 90 на 90 (диаметр нашей окружности), левый нижний и верхний правый углы определяются значениями 95=140-45 и 185=140+45. Оператор PUT при исполнении подряд два раза сначала рисует, а затем стирает изображение, таким образом создается иллюзия движения изображения. При запуске программы (85), как мы убедимся , движущаяся окружность проходит по неподвижной окружности, не искажая ее. Таким образом, мы достигли поставленной цели. Мы можем усложнить задачу, раскрасив окружности. С этой целью добавим в программу следующие строки:

 30 CIRCLE(220,220),45,15

 40 PAINT(140,140),4,15                                                            (86)

 50 PAINT(220,220),2,15

       Операторы GET и PUT позволяют перемещать по экрану и гораздо более сложные изображения, причем не искажая уже имеющиеся на экране изображения. В качестве достаточно простого примера заставим двигаться уже известное нам изображение двух сцепленных торов (Рис. 24). Соответствующая программа выглядит так:

 10 CLS

 20 X1=240:Y1=80:R=50:K=2

 30 B=55:C=410:D=3:E=1.5:R1=15

 40 GOSUB 100

 50 X1=270:Y1=120:K=1

 60 B=-60:C=263:D=4:E=.3:R1=9                                           

 70 GOSUB 100                                                                                              (87)

 80 GOTO 200

 100 FOR F=B TO C STEP D

 110 A=F*3.1416/180

 120 X2=X1+K*R*SIN(A)

 130 Y2=Y1+R*COS(A)

 140 CIRCLE(X2,Y2),R1,7,,,E                   

 150 NEXT F

 160 RETURN

 200 N=INT((230*2+7)/8)*160

 210 DIM TR(N)

 220 GET(125,175)-(355,15),TR 

 230 FOR  F=0 TO 354 STEP 6

 240 A=F*3.1416

 250 X=X1+60*SIN(A):Y=Y1+60*COS(A)                                                        

 260 PUT(X,Y),TR

 270 PUT(X,Y),TR

 280 NEXT

 290 GOTO 230

       Программа (87) по сути является соединением программы 82 (строки 10-160) и программы 85 (строки 200-290). Первая часть программы рисует на экране изображение двух сцепленных торов, строки 200-230 записывают прямоугольник, содержащий  это изображение, в массив TR, строки 230-290 вызывают вращение торов вокруг центра в точке X1,Y1 по окружности с радиусом, равным 60. Размеры прямоугольника с изображением торов подобраны после нескольких пробных попыток. При запуске программы мы  убедимся, что не происходит стирания неподвижного первоначального изображения и что движение происходит быстро, этого не удалось бы добиться при использовании приема последовательного рисования и стирания этого сложного изображения изменением его цвета от белого к черному. Это связано с очень быстрой работой оператора PUT при выводе сохраненного в массиве изображения на экран.

